

A SAS® Macro Implementation of a Test for Co-directional Interactions

Julia L. Sharp, Colorado State University; Patrick D. Gerard, Clemson University;
Bereket Tesfaldet, FDA

ABSTRACT

In two-factor factorial experiments, caution is often advised in interpreting the main effects when the interaction term is significant. However, when the interaction is co-directional relative to a particular factor, the main effect test for that factor can be considered. Gerard and Sharp (2011) developed a formal method for testing for co-directional interactions based on union-intersection and intersection-union techniques. A SAS® macro was developed for the implementation of this method in SAS. This presentation uses examples to demonstrate the method and the macro. The macro can be called within or outside of the GLM procedure, and it has four required and two optional input parameters. Depending on the user's request, the macro provides output that includes a brief ANOVA table that incorporates the test of the co-directional interaction for each factor or a detailed analysis including estimates of simple effects, corresponding standard errors, test statistics, and p-values for each 2x2 factorial arrangement of the two factors.

INTRODUCTION

In the analysis of two-factor factorial experimental design, an interaction effect is the combined effect of the factors on the dependent measure. When an interaction effect is statistically significant, the effect of one factor depends on the level of the other factor. In this case, we need to be cautious in testing and interpreting the main effects of each factor.

When an interaction is co-directional relative to a given factor, we may consider testing the main effect of the factor. However, when the interaction is not co-directional, testing the main effect of the factor may give misleading results. Gerard and Sharp (2011) developed a formal way of testing for co-directional interactions based on union–intersection (UI) and intersection–union (IU) methods. In this project, a SAS macro code is written to implement the method and examples are provided to demonstrate the macro code.

INTERACTIONS

Consider two-factor, say A and B, factorial design of experiment each with two levels, say A1 and A2 for factor A and B1 and B2 for factor B. The model for this experiment is given by

$$y_{ijk} = \mu_{ij} + \epsilon_{ijk}, \quad i = 1,2; j = 1,2; k = 1, \dots, n,$$

where y_{ijk} is the response of the k^{th} observation at the i^{th} level of factor A and j^{th} level of factor B, μ_{ij} is the mean response at the i^{th} level of factor A and j^{th} level of factor B, and ϵ_{ijk} are random errors which are assumed to be independent and normally distributed with mean 0 and variance σ^2 .

A simple effect of a factor is the change in the mean response of the factor as it changes from one level to a second level while the second factor is kept fixed at a given level. Hence, the simple effects of factor A at levels B1 and B2 of factor B are given, respectively, as $\mu_{11} - \mu_{21}$ and $\mu_{12} - \mu_{22}$. Similarly, the simple effects of factor B at levels A1 and A2 of factor A are given, respectively, as $\mu_{11} - \mu_{12}$ and $\mu_{21} - \mu_{22}$.

The average of the simple effects of a factor is called the main effect of the factor. For the 2x2 factorial design, the main effects of the factors A and B, respectively, are

$$\frac{1}{2}(\mu_{11} - \mu_{21}) + \frac{1}{2}(\mu_{12} - \mu_{22}) \quad (1)$$

and

$$\frac{1}{2}(\mu_{11} - \mu_{12}) + \frac{1}{2}(\mu_{21} - \mu_{22}). \quad (2)$$

The two factors A and B are said to be interacting when the effect of one of the factor on the response depends on the level of the other factor. To demonstrate interaction between factors graphically, we use interaction (also known as profile) plots. An interaction plot is a graph on a rectangular coordinate system that displays the levels of one factor on the horizontal axis and separate lines are provided for the means of each level of the other factor. The response variable is displayed on the vertical axis.

Figure 1 (a) shows an interaction plot where the effect of changing the level of factor A from A_1 to A_2 on the average of the response y is the same regardless of the level of factor B. Note that the lines are parallel indicating the absence of interaction in the factors. Thus, the underlying relationship between factor A and the response y is not affected by the level of factor B. In Figure 1 (b), the corresponding change in the average response does depend on the level of factor B showing an interaction between factors A and B. However, the change in the average response is in the same (positive) direction. Hinkelmann and Kempthorne (1994) referred to this kind of interaction as a co-directional. The interaction plot in Figure 1 (c) illustrates an interaction between the factors A and B. In this case, the change in the average response is in opposite direction. Hinkelmann and Kempthorne (1994) referred to this kind of interaction as anti-directional.

In analyzing a two-factor factorial design of experiments, we need to investigate whether a significant interaction between the two factors exists before testing the significance of the main effect of a factor. If a significant interaction exists, it is important to determine the nature of the interaction. In the presence of a co-directional interaction with regard to a given factor, a test of the main effect of the factor can be performed. However, if the interaction with regard to a given factor is not co-directional, the interaction between the factors may mask the main effect of the factor. Consequently, performing the test may give a misleading conclusion.




Figure 1 Interaction plot illustrating the absence of interaction in the factors (a); an interaction that is co-directional in nature (b); and an interaction that is anti-directional in nature (c)

TEST OF CO-DIRECTIONAL INTERACTION

Gerard and Sharp (2011) developed a formal way of testing the nature of interaction between two factors. The test is based on union–intersection (UI) and intersection–union (IU) methods. This section provides details about this test.

Algebraically, the interaction effect of two factors A and B is the average of the difference of the two simple effects of either factor A or B and is given by

$$\frac{1}{2}(\mu_{11} - \mu_{21}) - \frac{1}{2}(\mu_{12} - \mu_{22}). \quad (3)$$

Hence, to determine the existence of a significant interaction, we test the hypothesis

$$H_o: \mu_{11} - \mu_{12} - \mu_{21} + \mu_{22} = 0. \quad (4)$$

When the null hypothesis is true, the simple effects of factor A are equal, and thus, $\mu_{11} - \mu_{21} = \mu_{12} - \mu_{22}$. Similarly, $\mu_{11} - \mu_{12} = \mu_{21} - \mu_{22}$ which indicates the simple effects of B are equal. Thus, when there is not an interaction, the simple effects of each factor are equal. On the other hand, when the null hypothesis (4) is not true, the simple effects of each factor are not equal indicating an interaction is present. If the interaction between factors A and B is co-directional with regard to factor A, then the orientation of the simple effects of A must be the same. That is, either

$$\mu_{11} - \mu_{21} > 0 \text{ and } \mu_{12} - \mu_{22} > 0 \quad (5)$$

or

$$\mu_{11} - \mu_{21} < 0 \text{ and } \mu_{12} - \mu_{22} < 0. \quad (6)$$

Defining the hypotheses H1 to H4 as

$$\begin{aligned} H1: \mu_{11} - \mu_{21} &\geq 0 \\ H2: \mu_{12} - \mu_{22} &\geq 0 \\ H3: \mu_{11} - \mu_{21} &\leq 0, \text{ and} \\ H4: \mu_{12} - \mu_{22} &\leq 0 \end{aligned} \quad (7)$$

the hypothesis

H_o : Interaction is not co-directional for factor A

H_a : Interaction is co-directional for factor A

can be written as

$$\begin{aligned} H_o: (H1 \cup H2) \cap (H3 \cup H4) \\ H_a: (\sim H1 \cap \sim H2) \cup (\sim H3 \cap \sim H4) \end{aligned} \quad (8)$$

where $\sim Y$ represents the complement of Y. Following the IU method, the null hypothesis H_o is rejected with significance level α if either $H1 \cup H2$ or $H3 \cup H4$ is rejected at level of significance α . Using the UI method, $H1 \cup H2$ is rejected with significance level α if both H1 and H2 are rejected each at α level of significance. Similarly, $H3 \cup H4$ is rejected with significance level α if both H3 and H4 are rejected each at level of significance α . The combination of these two methods is called IU–UI (or UI–IU) method. Thus, we can test the hypothesis (8) by testing the four hypotheses H1 to H4 each at level of significance α .

Since each hypothesis in (8) is a simple contrast between cell means, each test can be carried out using a one-tailed Student's t-test.

To extend the above test of co-directional interaction beyond the 2^2 factorial design, suppose factor A has a levels and factor B has b levels. An interaction between A and B is said to be co-directional if all possible 2×2 factorial arrangements involving any two levels of factor A and any two levels of factor B are co-directional. Note that there are $\binom{a}{2} \binom{b}{2}$ factorial arrangements to be considered. Let

$$L = \{ (p, q) \times (u, v) : 1 \leq p < q \leq a, 1 \leq u < v \leq b \}$$

be the set of all such 2×2 factorial arrangements. The null hypothesis that interaction is not co-directional with respect to factor A can be written as

$$H_0 : \bigcup_{(p,q) \times (u,v) \in L} \text{Interaction is not co-directional with respect to factor A.} \quad (9)$$

This is an UI hypothesis, and thus, would be rejected at α level of significance only if all the hypotheses associated with the elements in L are rejected each at α level of significance. Gerard and Sharp (2011) showed that the overall Type-I error rates of testing the null hypotheses in (8) and in (9) are maintained at the nominal level α .

TEST OF CO-DIRECTIONAL INTERACTION MACRO

The UI-IU method of testing co-directional interaction is implemented using a SAS macro called CDTTEST. CD stands for co-directional and TTEST stands for Student's t-test. The code for the CDTTEST macro is provided in the Appendix. The CDTTEST macro can be called within or outside of PROC GLM and has four required and two optional parameters.

The required input parameters are:

- `mydata` = The name of the SAS data set for the analysis.
- `var` = The name of the dependent variable in the model.
- `factor1` = The name of the first factor (A) in the model.
- `factor2` = The name of the second factor (B) in the model.

The optional input parameters are:

- `cdalpha` = The level of significance for testing co-directional interaction. The value must be between 0 and 1. If not specified, a default value of 0.05 is used.
- `Summary` = The level of summary output required by the user. If not specified, a default value of 'a' is used.
 - `'a'` = The output is the ANOVA table with the result of the co-directional interaction test for each factor.
 - `'b'` = The output is a brief summary table with p-values for each possible simple effect in addition to the ANOVA table given by option 'a'.
 - `'d'` = The output is a detailed table containing estimates of all possible simple effects and their corresponding standard errors, t-statistics and p-values in addition to the tables provided by option 'b'.

The SAS code to call the CDTTEST macro within PROC GLM is shown below:

```

/*****
/* cdttest macro implemented within proc glm with data set twofactor dat,*/
/* response variable Y, factors A and B, using a significance level of */
/* 0.05, and obtaining the co-directional interaction test results with */
/* the ANOVA table */
/*****
proc glm data=twofactordat;
class A B;
  model Y=A B A*B;
  %cdttest(mydata=twofactordat,
          var=Y,
          factor1=A,
          factor2=B,
          calpha=0.05,
          summary='a');
run;
quit;

```

EXAMPLE APPLICATIONS OF THE CO-DIRECTIONAL INTERACTION TEST MACRO

The CDTTEST macro is applied to the three examples presented in Gerard and Sharp (2011).

EXAMPLE 1: TWO FACTORS WITH TWO LEVELS (CO-DIRECTIONAL INTERACTION IN ONE VARIABLE BUT NOT THE OTHER)

Two light treatment times (10 hour exposure and 12 hour exposure; factor A) and two temperatures (70°F and 80°F; factor B) and were studied in an example from Rao (1998). The implementation of the CDTTEST macro is given as:

```

%INCLUDE 'CDTTEST.sas';

DATA mydata;
  INPUT Time $ Temp $ Weight @@;
  DATALINES;
  10 70 10.6 10 70 9.8 10 70 10.1 10 70 10.7
  10 80 18.3 10 80 20.1 10 80 19.6 10 80 20.2
  12 70 11.3 12 70 10.9 12 70 10.5 12 70 10.2
  12 80 43.2 12 80 33.8 12 80 34.6 12 80 33.5
  ;
RUN;

PROC GLM DATA=mydata;
  CLASS Time Temp;
  MODEL Weight = Time Temp Time*Temp;
  %CDTTEST(mydata=mydata, var=Weight, factor1=Time, factor2=Temp,
          summary='d');
RUN;

```

The output from the CDTTEST macro using the `summary='d'` option includes the tests of the simple effects, a brief summary table of hypotheses H1 through H4, and the ANOVA table from PROC GLM with a summary of whether the interactions are co-directional or not co-directional (Tables 1-3). The estimates of the simple effects, the standard errors of the estimates, and test statistics and p-values associated with testing the associated direction of the simple effects are given in Table 1. For example, for testing H3: $\mu_{11} - \mu_{21} \leq 0$, the simple effect for time at temperature 70, the estimate of the simple effect is -0.425

with standard error 1.685. The test statistic is $t = -0.252$, $p = 0.403$. For testing $H_4: \mu_{12} - \mu_{22} \leq 0$, the simple effect for time at temperature 80, the test statistic is -9.93 , $p < 0.001$. A brief summary of the p-values given in Table 1 is provided in Table 2, along with a conclusion for the test of co-directional interaction. For example, the interaction with respect to time is not co-directional because both H_3 and H_4 are not rejected ($p = 0.403$ and $p < 0.001$, respectively). The default output from the CDTTEST macro is the ANOVA table with the usual degrees of freedom, sums of squares, means squares, F-test statistic, and p-value, along with the result of the test for co-directional interaction with respect to each factor (Table 3).

Tests of Co-directional Interactions using UI-IU Methods

Obs	Factor	Level	Hypothesis	Estimate	Stderr	TValue	ProbT
1	Time	(10, 12) x (70, 80)	SE of Time at Temp=70 >=	-0.425	1.68455	-0.2523	0.59746
2	Time	(10, 12) x (70, 80)	SE of Time at Temp=80 >=	-16.725	1.68455	-9.9285	1.00000
3	Time	(10, 12) x (70, 80)	SE of Time at Temp=70 <=	-0.425	1.68455	-0.2523	0.40254
4	Time	(10, 12) x (70, 80)	SE of Time at Temp=80 <=	-16.725	1.68455	-9.9285	0.00000
5	Temp	(10, 12) x (70, 80)	SE of Temp at Time=10 >=	-9.250	1.68455	-5.4911	0.99993
6	Temp	(10, 12) x (70, 80)	SE of Temp at Time=12 >=	-25.550	1.68455	-15.1673	1.00000
7	Temp	(10, 12) x (70, 80)	SE of Temp at Time=10 <=	-9.250	1.68455	-5.4911	0.00007
8	Temp	(10, 12) x (70, 80)	SE of Temp at Time=12 <=	-25.550	1.68455	-15.1673	0.00000

Table 1 Estimates of simple effects, standard errors, test statistics, and p-values for testing the direction of the simple effect

**Brief Summary of
Tests of Co-directional Interactions using UI-IU Methods
(at Alpha 0.05)**

Obs	Factor	Level	H1	H2	H3	H4	Conclusion
1	Time	(10, 12) x (70, 80)	0.5975	1.0000	0.4025	0.0000	Not Co-Dir
2	Temp	(10, 12) x (70, 80)	0.9999	1.0000	0.0001	0.0000	Co-Dir

Table 2 Brief summary table of p-values for consideration of each possible simple effect including a conclusion to the test of co-directional interaction with respect to each factor

**Analysis of Variance Table with UI-IU Method
of Testing Co-directional Interactions
(at Alpha 0.05)**

Obs	Source	Interaction	DF	SS	MS	FValue	ProbF
1	Time	Not Co-Dir	1	294.122500	294.122500	51.82	<.0001
2	Temp	Co-Dir	1	1211.040000	1211.040000	213.38	<.0001
3	Time*Temp		1	265.690000	265.690000	46.81	<.0001

Table 3 ANOVA table with the result of the test for co-directional interaction for each factor, time and temperature

EXAMPLE 2: TWO FACTORS WITH TWO LEVELS (INTERACTION NOT CO-DIRECTIONAL FOR EITHER FACTOR)

Example 2 is from Cody and Smith (2006). Activity levels (gp, factor A) were measured for both normal and hyperactive children. Some of the children had been given Ritalin and some had been given a placebo (trt, factor B). There were four children in each of the two factor combinations (16 total children). The implementation of the CDTTEST macro is given as:

```
%INCLUDE 'CDTTEST.sas';

DATA one;
  INPUT gp $ trt $ act @@;
  CARDS;
  n p 50 n p 45 n p 55 n p 52 n r 67
  n r 60 n r 58 n r 65 h p 70 h p 72
  h p 68 h p 75 h r 51 h r 57 h r 48
  h r 55
  ;
RUN;

PROC GLM DATA = one;
  CLASS gp trt;
  MODEL act=gp trt gp*trt;
  %CDTTEST(mydata=one, var=act, factor1=gp, factor2=trt,
           summary='d');
RUN;
```

The output from the CDTTEST macro using the summary='d' option includes the tests of the simple effects, a brief summary table of hypotheses H1 through H4, and the ANOVA table from PROC GLM with a summary of whether the interactions are co-directional or not co-directional (Tables 4-6). The estimates of the simple effects, the standard errors of the estimates, and test statistics and p-values associated with testing the associated direction of the simple effects are given in Table 4. For example, for testing H3: $\mu_{11} - \mu_{21} \leq 0$, the simple effect for group for the placebo treatment, the estimate of the simple effect is 20.75 with standard error 2.75. The test statistic is $t = 7.55$, $p=1.00$. For testing H4: $\mu_{12} - \mu_{22} \leq 0$, the simple effect group for the Ritalin treatment, the test statistic is -3.55, $p=0.002$. A brief summary of the p-values given in Table 4 is provided in Table 5, along with a conclusion for the test of co-directional interaction. For example, the interaction with respect to group is not co-directional because both H3 and H4 are not rejected ($p=1.00$ and $p=0.002$, respectively). The default output from the CDTTEST macro is the ANOVA table with the usual degrees of freedom, sums of squares, means squares, F-test statistic, and p-value, along with the result of the test for co-directional interaction with respect to each factor (Table 6).

Tests of Co-directional Interactions using UI-IU Methods

Obs	Factor	Level	Hypothesis	Estimate	Stderr	TValue	ProbT
1	gp	(h, n) x (p, r)	SE of gp at trt=p >=	20.75	2.75	7.54545	0.00000
2	gp	(h, n) x (p, r)	SE of gp at trt=r >=	-9.75	2.75	-3.54545	0.99798
3	gp	(h, n) x (p, r)	SE of gp at trt=p <=	20.75	2.75	7.54545	1.00000
4	gp	(h, n) x (p, r)	SE of gp at trt=r <=	-9.75	2.75	-3.54545	0.00202
5	trt	(h, n) x (p, r)	SE of trt at gp=h >=	18.50	2.75	6.72727	0.00001
6	trt	(h, n) x (p, r)	SE of trt at gp=n >=	-12.00	2.75	-4.36364	0.99954
7	trt	(h, n) x (p, r)	SE of trt at gp=h <=	18.50	2.75	6.72727	0.99999
8	trt	(h, n) x (p, r)	SE of trt at gp=n <=	-12.00	2.75	-4.36364	0.00046

Table 4 Estimates of simple effects, standard errors, test statistics, and p-values for testing the direction of the simple effect

**Brief Summary of
Tests of Co-directional Interactions using UI-IU Methods
(at Alpha 0.05)**

Obs	Factor	Level	H1	H2	H3	H4	Conclusion
1	gp	(h, n) x (p, r)	0.0000	0.9980	1.0000	0.0020	Not Co-Dir
2	trt	(h, n) x (p, r)	0.0000	0.9995	1.0000	0.0005	Not Co-Dir

Table 5 Brief summary table of p-values for consideration of each possible simple effect including a conclusion to the test of co-directional interaction with respect to each factor

**Analysis of Variance Table with UI-IU Method
of Testing Co-directional Interactions
(at Alpha 0.05)**

Obs	Source	Interaction	DF	SS	MS	FValue	ProbF
1	gp	Not Co-Dir	1	121.0000000	121.0000000	8.00	0.0152
2	trt	Not Co-Dir	1	42.2500000	42.2500000	2.79	0.1205
3	gp*trt		1	930.2500000	930.2500000	61.50	<.0001

Table 6 ANOVA table with the result of the test for co-directional interaction for each factor, gp and trt

EXAMPLE 3: TWO FACTORS, ONE FACTOR WITH THREE LEVELS AND THE OTHER FACTOR WITH TWO LEVELS

This example is modified from Ott and Longnecker (2001). Yield of soybeans was compared at three levels of copper (factor A, 1, 3, 7) and two levels of manganese (factor B, 20, 110). The implementation of the CDTTEST macro is given as:

```
%INCLUDE 'CDTTEST.sas';

DATA one;
  INPUT cu $ mn $ yield @@;
  DATALINES;
  1 20 1558 1 20 1578 1 110 2830 1 110 2810
  3 20 1590 3 20 1610 3 110 2860 3 110 2841
  7 20 1328 7 20 1427 7 110 2960 7 110 2941
  ;
RUN;

PROC GLM;
  CLASS cu mn;
  MODEL yield=cu mn cu*mn;
  %CDTTEST(mydata=one, var=yield, factor1=cu, factor2=mn,
           summary='b');
RUN;
```

The output from the CDTTEST macro using the `summary='b'` option includes a brief summary table of hypotheses H1 through H4 and the ANOVA table from PROC GLM with a summary of whether the interactions are co-directional or not co-directional (Tables 7-8). A brief summary of the p-values is provided in Table 7, along with a conclusion for the test of co-directional interaction. For example, the interaction with respect to copper (cu) is not co-directional but the interaction with respect to manganese is co-directional. The default output from the CDTTEST macro is the ANOVA table with the usual degrees of freedom, sums of squares, means squares, F-test statistic, and p-value, along with the result of the test for co-directional interaction with respect to each factor (Table 8).

**Brief Summary of
Tests of Co-directional Interactions using UI-IU Methods
(at Alpha 0.05)**

Obs	Factor	Level	H1	H2	H3	H4	Conclusion
1	cu	(1, 3) x (110, 20)	0.8166	0.8273	0.1834	0.1727	Not Co-Dir
2	cu	(1, 7) x (110, 20)	0.9971	0.0004	0.0029	0.9996	Not Co-Dir
3	cu	(3, 7) x (110, 20)	0.9907	0.0002	0.0093	0.9998	Not Co-Dir
4	mn	(1, 3) x (110, 20)	0.0000	0.0000	1.0000	1.0000	Co-Dir
5	mn	(1, 7) x (110, 20)	0.0000	0.0000	1.0000	1.0000	Co-Dir
6	mn	(3, 7) x (110, 20)	0.0000	0.0000	1.0000	1.0000	Co-Dir

Table 7 Brief summary table of p-values for consideration of each possible simple effect including a conclusion to the test of co-directional interaction with respect to each factor

**Analysis of Variance Table with UI-IU Method
of Testing Co-directional Interactions
(at Alpha 0.05)**

Obs	Source	Interaction	DF	SS	MS	FValue	ProbF
1	cu	Not Co-Dir	2	7504.167	3752.083	3.84	0.0843
2	mn	Co-Dir	1	5536566.750	5536566.750	5667.39	<.0001
3	cu*mn		2	69016.500	34508.250	35.32	0.0005

Table 8 ANOVA table with the result of the test for co-directional interaction for each factor, cu and mn

CONCLUSION

The CDTTEST macro is an easy to implement tool for considering the nature of interactions between two factors. The macro produces three levels of reports: 1) a report of the test of the simple effects for factors in both directions (positive and negative); 2) a brief summary of the p-values for the tests of the simple effects for factors in both directions (positive and negative); and 3) the ANOVA table from PROC GLM including a summary conclusion of the test for the co-directional interaction. These reports have been illustrated using the examples in Gerard and Sharp (2011). This macro can be used in conjunction with interaction plots to help describe the nature of the interaction between two factors.

REFERENCES

- Cody, R. P. and Smith, J. K. 2006. *Applied Statistics and SAS® Programming Language*, Upper Saddle River, NJ: Pearson Prentice-Hall.
- Gerard, P. D. and Sharp, J. L. 2011. "Testing for co-directional interactions using union–intersection and intersection–union methods." *Journal of Applied Statistics*, 38(7): 1347-1358.
- Hinkelmann, K. and Kempthorne, O. 1994. *Design and Analysis of Experiments, Volume 1: Introduction to Experimental Design*. NY: Wiley.
- Ott, R. L. and M. Longnecker, M. 2001. *An Introduction to Statistical Methods and Data Analysis*, 5th ed., Pacific Grove, CA: Duxbury.
- Rao, P. V. 1998. *Statistical Research Methods in the Life Sciences*, Pacific Grove, CA: Brooks/Cole.

APPENDIX: SOURCE CODE OF CO-DIRECTIONAL INTERACTION TEST (CDTTEST) MACRO

/*****

Description:

CDTTEST is a SAS macro that tests for co-direction interaction using Union-Intersection and Intersection-Union methods in a two-factor factorial design of experiment.

Usage:

```
%CDTTEST(mydata=DD, var=Y, factor1=A, factor2=B, cdalpha=0.05,
          summary='a')
%CDTTEST(mydata=DD, var=Y, factor1=A, factor2=B, cdalpha=0.05)
%CDTTEST(mydata=DD, var=Y, factor1=A, factor2=B, summary='a')
%CDTTEST(mydata=DD, var=Y, factor1=A, factor2=B)
```

Required Parameters:

mydata=DD:	Specifies the dataset for the analysis.
Var = Y:	Specifies the dependent variable in the model.
factor1 = A:	Specifies the first factor in the model.
factor2 = B:	Specifies the second factor in the model.

Optional Parameters:

cdalpha = p:	Specifies the level of significance p for testing co-directional interaction. The value must be between 0 and 1. If not specified, a default value of 0.05 is used.
summary='a':	Specifies the level of summary output required by the user. If not specified, a default value of 'a' is used.
'a' =	ANOVA table with the result of test for co-directional interaction on each factor.
'b' =	Brief summary table with p-values for each possible simple effect in addition to the ANOVA table given by option 'a'.
'd' =	Detailed table containing estimates of all possible simple effects and their corresponding standard errors, t statistics and p-values in addition to the tables provided by option 'b'.

Reference:

Patrick D. Gerard & Julia L. Sharp (2011). Testing for co-directional interactions using union-intersection and intersection-union methods, *Journal of Applied Statistics*, 38:7, 1347-1358.

*****/

PROC FORMAT;

```
VALUE status 1 = 'Not Co-Dir'
              0 = 'Co-Dir';
```

RUN;

```

%MACRO CDTTEST(mydata=, var=, factor1=, factor2=, cdalpha=0.05,
               summary='a');
  * Use PROC GLM to generate the dataset to work with;
  PROC GLM;
    ODS SELECT NONE;
    ODS OUTPUT overallanova = glmoverallanova;
    ODS OUTPUT 'Type III Model ANOVA' = glmanova;
    CLASS &factor1 &factor2;
    MODEL &var = &factor1 &factor2 &factor1*&factor2;
    LSMEANS &factor1*&factor2 /OUT = glmlsmeans;
  RUN;
  ODS SELECT ALL;

  * Extract the MSE, DF and N from the ANOVA table and store them in
  macro variables;
  DATA _NULL_;
    SET glmoverallanova;
    IF Source = 'Error' THEN DO;
      CALL SYMPUTX('mse', MS);
      CALL SYMPUTX('edf', df);
    END;
    ELSE IF Source = 'Corrected Total' THEN
      CALL SYMPUTX('nobs', df+1);
  RUN;

  * Modify the glmlsmeans dataset;
  PROC FREQ DATA=&mydata;
    ODS SELECT NONE;
    TABLE &factor1*&factor2/NOCUM nopercnt list OUT=replications;
  RUN;
  ODS SELECT ALL;

  DATA glmlsmeans;
    MERGE glmlsmeans(DROP=_NAME_ DROP=STDERR)
          replications(DROP=PERCENT);
    BY &factor1;
  RUN;

```

```

* Identifies distinct levels of the factors;
PROC SQL NOPRINT;
    SELECT COUNT(DISTINCT &factor1)
        INTO :num_lev_&factor1
        FROM glmmeans;
    %LET num_lev_&factor1 = &&num_lev_&factor1;

    SELECT DISTINCT &factor1
        INTO :&factor1._lev1-:&factor1._lev&&num_lev_&factor1
        FROM glmmeans;

    SELECT COUNT(DISTINCT &factor2)
        INTO :num_lev_&factor2
        FROM glmmeans;
    %LET num_lev_&factor2 = &&num_lev_&factor2;
    SELECT DISTINCT &factor2
        INTO :&factor2._lev1-:&factor2._lev&&num_lev_&factor2
        FROM glmmeans;
QUIT;

* Generates simple effect hypothesis for a given 2x2 factorial
arrangement;
%MACRO CELLS(factor=, A_lev1=, A_lev2=, B_lev1=, B_lev2=);

    %LOCAL i;
    %LOCAL j;
    %DO i=1 %TO 2;
        %DO j=1 %TO 2;
            DATA _NULL_;
                SET glmmeans;
                WHERE &factor1="&&A_lev&i" AND
                    &factor2="&&B_lev&j";
                CALL SYMPUTX("cell&i&j", LSMEAN);
                CALL SYMPUTX("size&i&j", COUNT);
            RUN;
        %END;
    %END;

    %IF &factor EQ 1 %THEN %DO;
        %LET mf = &factor;
        %LET sf = 2;
        %LET Lf = B;
        %LET CR1 = 21;
        %LET CR2 = 12;
    %END;
    %ELSE %DO;
        %LET mf = &factor;
        %LET sf = 1;
        %LET Lf = A;
        %LET CR1 = 12;
        %LET CR2 = 21;
    %END;
    %LET inequality = %STR(>=, <=);
    %LOCAL h;

```

```

* Creates a table for co-directional test for each 2x2
  factorial arrangement of a given factor;

DATA cdtable&t;
  LENGTH Factor $ 15 Level $ 30 Hypothesis $ 30 ;
  Factor = "&&factor&mf";
  Level = "(&A_lev1, &A_lev2) x (&B_lev1, &B_lev2)";
  %DO k=1 %TO 2;
    %DO h=1 %TO 2;
      Hypothesis = "SE of &&factor&mf at
                    &&factor&sf=&&&Lf._lev&h
                    %SCAN(&inequality, &k, %STR(,))";
      Estimate = ((-1)**%EVAL(&h+1))*(&&cell&h&h -
                    &&&&cell&&CR&h);
      Stderr = sqrt(&mse * (1/&&size&h&h +
                    1/&&&&size&&CR&h));
      TValue = Estimate/STDERR;
      ProbtT = %EVAL(2-&k) +
              ((-1)**%EVAL(&k))*PROBT(TValue, &edf);
      IF ProbtT < &cdalpha THEN Conclusion = 0;
      ELSE Conclusion = 1;
      OUTPUT;
    %END;
  %END;
RUN;

* Creates a summary table for co-directional test for each 2x2
  factorial arrangement of a given factor;
PROC SQL NOPRINT;
  SELECT ProbtT, Conclusion
  INTO :P1-:P4, :H1-:H4
  FROM cdtable&t;

  CREATE TABLE summary&t as
  SELECT DISTINCT Factor, Level, &P1 as H1 format 8.4,
  &P2 as H2 format 8.4, &P3 as H3 format 8.4, &P4 as H4
  format 8.4, %EVAL((&H1 OR &H2) AND (&H3 OR &H4)) as
  Conclusion FROM cdtable&t;
QUIT;

%MEND CELLS;

```

```

* Carries out co-directional interaction test for each 2x2 factorial
  arrangement of all factors;
%LOCAL f;
%LET t = 1;
%LOCAL i;
%LOCAL j;
%LOCAL p;
%LOCAL q;
%DO f=1 %TO 2; /*      For each factor */
  %DO i=1 %TO &&num_lev_&factor1;
    %DO j=%EVAL(&i+1) %TO &&num_lev_&factor1;
      %DO p=1 %TO &&num_lev_&factor2;
        %DO q=%EVAL(&p+1) %TO &&num_lev_&factor2;

          %CELLS(factor=&f,
A_lev1=&&&factor1._lev&i, A_lev2=&&&factor1._lev&j,
B_lev1=&&&factor2._lev&p, B_lev2=&&&factor2._lev&q);

          %Let t = %EVAL(&t + 1);
        %END;
      %END;
    %END;
  %END;
%END;

* Combine subtables into tables;
%LOCAL m;
%DO m=2 %TO %EVAL(&t - 1);
  PROC APPEND DATA=cetable&m OUT=cetable1;
  RUN;
  PROC APPEND DATA=summary&m OUT=summary1;
  RUN;
%END;

*Create summary tables;
PROC SQL NOPRINT;
  CREATE TABLE summary as
  SELECT Factor,
    CASE
      WHEN MAX(Conclusion) = 1 THEN 'Not Co-Dir'
      ELSE 'Co-Dir'
    END as Interaction
  FROM summary1 GROUP BY Factor;

  CREATE TABLE cdanova_temp as
  SELECT G.Source, S.Interaction, G.DF, G.SS, G.MS, G.FValue,
  G.ProbF FROM glmanova as G left join summary as S
  on G.Source = S.Factor;

```

```

CREATE TABLE cdanova as
SELECT Source, Interaction, DF, SS, MS, FValue, ProbF,
CASE
    WHEN Source = "&factor1" THEN 1
    WHEN Source = "&factor2" THEN 2
    ELSE 3
END as Sorter
FROM cdanova_temp
ORDER BY Sorter;
QUIT;

* Print out summary tables;
%IF &summary = 'd' %THEN %DO;
    PROC PRINT DATA=chtable1;
        VAR Factor Level Hypothesis Estimate Stderr TValue ProbT;

        TITLE "Tests of Co-directional Interactions using UI-IU
        Methods";

    RUN;
%END;

%IF &summary = 'b' OR &summary = 'd' %THEN %DO;
PROC PRINT DATA=summary1;
    VAR Factor Level H1 H2 H3 H4 Conclusion;
    FORMAT Conclusion status.;
    TITLE "Brief Summary of";
    TITLE2 "Tests of Co-directional Interactions using UI-IU
    Methods";
    TITLE3 "(at Alpha &cdalpha)";
RUN;
%END;

PROC PRINT DATA=cdanova;
    VAR Source Interaction DF SS MS FValue ProbF;
    TITLE "Analysis of Variance Table with UI-IU Method";
    TITLE2 "of Testing Co-directional Interactions";
    TITLE3 "(at Alpha &cdalpha)";
RUN;

%MEND CDTTEST;

```

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Julia L. Sharp
Colorado State University
julia.sharp@colostate.edu